
POLY234 
Display 

  

(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020) 

 

Overview1 

Program POLY234 program calculates (!) the roots of 2nd, 3rd and 4th  degree polynomials. Both real and 
complex roots can be calculated on the following basis: 

 Degree  Polynomial expression  Method of calculation 

 2nd  f(x) = x2 + a1x + a0  Quadratic formula 

 3rd  f(x) = x3 + a2x2 + a1x + a0  Cardano's algorithm 

 4th  f(x) = x4 + a3x3+ a2x2 + a1x + a0  Ferrari's algorithm 

 

 

Another method of finding the roots of polynomials is through the Newton-Raphson method which is not 
part of this program. The reason being that finding roots by convergence is not deterministic and requires 
a different approach to programming.  

Please note my default FIX 5 and European (flag 28) setting in below examples. 

 

Examples: 2nd degree polynomials 

Find the (real) roots of the following polynomial: f(x) = x2 + 2x - 15 

 
1 This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no 
liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof 

https://www.swissmicros.com/

KEYSTROKES DISPLAY COMMENTS 

  Run POLY234 for 2nd degree 

[XEQ] [ALPHA]POLY234[ALPHA]  Enter degree of the polynomial 

2[R/S]  Enter coefficients a0=-15 and a1=2 

-15[ENTER]2[R/S]  Root 1 (in Y register) 

[R/S]  Root 2 (in Y register) 

 

Find the (complex, based on i2=-1) roots of the following polynomial: f(x) = x2 + 2x + 15 

KEYSTROKES DISPLAY COMMENTS 

[R/S]   Run POLY234 again 

[R/S]  Enter degree of the polynomial 

2[R/S]  Enter coefficients a0=15 and a1=2 

15[ENTER]2[R/S]  Root 1 (in X and Y register) 

[R/S]  Root 2 (in X and Y register) 

 

Find the (complex) roots of the following polynomial: f(x) = x2 + 144 

KEYSTROKES DISPLAY COMMENTS 

[R/S]  Run POLY234 again 

[R/S]  Enter degree of the polynomial 

2[R/S]  Enter coefficients a0=144 and a1=0 

144[ENTER]0[R/S]  Root 1 (in X register) 

[R/S]  Root 2 (in X register) 

 

 

Examples: 3rd degree polynomials 

Find the (real) roots of the following polynomial: f(x) = x3 - 37x2 -205x + 12121 

KEYSTROKES DISPLAY COMMENTS 

  Run POLY234 for 3rd degree 

[XEQ] [ALPHA]POLY234[ALPHA]  Enter degree of the polynomial 

3[R/S]  Enter a0=12121, a1=-205, a2=-37 

12121[ENTER]-205[ENTER]-37   

[R/S]  Root 1 (in Y register) 

[R/S]  Root 2 (in Y register) 

[R/S]  Root 3 (in Y register) 

 

 

Find the (complex) roots of the following polynomial: f(x) = x3 - 10x2 + x + 150 

KEYSTROKES DISPLAY COMMENTS 

[R/S]  Run POLY234 for 3rd degree 

[R/S]  Enter degree of the polynomial 

3[R/S]  Enter a0=150, a1=1, a2=-10 

150[ENTER]1[ENTER]-10   

[R/S]  Root 1 (in X and Y register) 

[R/S]  Root 2 (in Y register) 

[R/S]  Root 3 (in X and Y register) 
  

 

Examples: 4th degree polynomials 

Find the (real) roots of the following polynomial: f(x) = x4 - 11x3 + x2 + 279x - 630 

KEYSTROKES DISPLAY COMMENTS 

  Run POLY234 for 4th degree 

[XEQ] [ALPHA]POLY234[ALPHA]  Enter degree of the polynomial 

4[R/S]  Enter a0=-630, a1=279, a2=1, a3=-11 

-630[ENTER]279[ENTER] 

1[ENTER]-11 
  

[R/S]  Root 1 (in Y register) 

[R/S]  Root 2 (in Y register) 

[R/S]  Root 3 (in Y register) 

[R/S]  Root 4 (in Y register) 

 

Find the (complex) roots of the following polynomial: f(x) = x4 + 10x3 - 26x2 + 251x + 540 

KEYSTROKES DISPLAY COMMENTS 

[R/S]  Run POLY234 for 4th degree 

[R/S]  Enter degree of the polynomial 

4[R/S]  Enter a0=540, a1=251, a2=-26, a3=10 

540[ENTER]251[ENTER] 

-26[ENTER]10 
  

[R/S]  Root 1 (in Y register) 

[R/S]  Root 2 (in Y register) 

[R/S]  Root 3 (in X and Y register) 

[R/S]  Root 4 (in X and Y register) 

[R/S]  Complex root 4 part in X register 

[X<>Y]  Real root 4 part in Y register 

[R/S]  Run POLY234 for another polynomial 

 

 

Program Listing 

The listing of program POLY234 is given on the next page.  The FIX 7 step in line 362 can sometimes 
lead to erroneous complex roots due to numerical limitations of the calculator's mantissa. You may have 
to change it to FIX 6 when working with small and large coefficient values. 

 01▪LBL "POLY234" 

 02 CF 02 

 03 CF 03 

 04 CF 04 

 05 "DEGREE=?" 

 06 PROMPT 

 07 "a0,a1" 

 08 GTO IND X 

 09▪LBL 03 

 10▪LBL 04 
 11 >",a2" 

 12 GTO IND X 

 13▪LBL 04 

 14 >",a3=?" 

 15 PROMPT 

 16 STO 14 
 17 X^2 

 18 X<>Y 

 19 STO 05 

 20 * 

 21 16 

 22 / 

 23 X<>Y 

 24 STO 06 

 25 RCL 14 

 26 * 

 27 4 

 28 / 
 29 - 

 30 + 

 31 RCL 14 

 32 X^2 

 33 RCL X 

 34 8 

 35 / 

 36 STO 07 

 37 * 

 38 3 

 39 * 

 40 32 
 41 / 

 42 - 

 43 STO 13 

 44 RCL 06 

 45 RCL 14 

 46 RCL 05 
 47 * 

 48 2 

 49 / 

 50 - 

 51 RCL 07 

 52 RCL 14 

 53 * 

 54 + 

 55 2 

 56 / 

 57 STO 12 

 58 X=0? 

 59 SF 03 

 60 X<0? 

 61 SF 02 

 62 RCL 05 
 63 RCL 07 

 64 3 

 65 * 

 66 - 

 67 STO 11 

 68 4 
 69 ST/ 14 

 70 FC? 03 

 71 GTO 11 

 72 RCL 13 

 73 X<>Y 

 74 X<> Z 

 75 XEQ 09 

 76 CF 03 

 77 SF 04 

 78 CHS 

 79 . 

 80 STO 10 
 81 XEQ 09 

 82 CLST 

 83 RCL 07 

 84 CHS 

 85 . 

 86 GTO 09 

 87▪LBL 11 

 88 RDN 

 89 2 

 90 / 

 91 STO Y 

 92 RCL 13 
 93 * 

 94 RCL 12 

 95 X^2 

 96 2 

 97 / 

 98 - 
 99 RCL 13 

100 CHS 

101 RCL Z 

102 CHS 

103 SF 04 

104 XEQ 10 

105 15 

106 STO 06 

107 RCL 12 

108 ABS 

109▪LBL 14 

110 2 

111 RCL 01 

112 ST- IND 06 

113 RDN 

114 RCL IND 06 
115 * 

116 LASTX 

117 X^2 

118 RCL 13 

119 - 

120 X<0? 
121 GTO 12 

122 SQRT 

123 STO 07 

124 X<>Y 

125 RCL 11 

126 - 

127 X<0? 

128 GTO 12 

129 SQRT 

130 FS?C 02 

131 CHS 

132 STO 08 
133 * 

134 ABS 

135 X<>Y 

136 - 

137 ABS 

138 .1 

139 X>Y? 

140 GTO 13 

141 LASTX 

142 STO Z 

143▪LBL 12 

144 RCL Z 
145 1 

146 ST+ 06 

147 RDN 

148 GTO 14 

149▪LBL 13 

150 . 
151 RCL IND 06 

152 RCL 07 

153 - 

154 RCL 08 

155 XEQ 09 

156 RCL IND 06 

157 RCL 07 

158 + 

159 RCL 08 

160 CHS 

161 GTO 09 

162▪LBL 02 

163 CLST 

164 STO 10 

165 >"=?" 

166 PROMPT 
167▪LBL 09 

168 CF 01 

169 2 

170 / 

171 CHS 

172 STO 00 
173 X^2 

174 X<>Y 

175 - 

176 X<0? 

177 SF 01 

178 ABS 

179 SQRT 

180 STO 01 

181 FC? 04 

182 GTO 02 

183 RCL 14 

184 ST- 00 
185 RDN 

186▪LBL 02 

187 RCL 00 

188 FC? 01 

189 + 

190 FS? 03 

191 STO 07 

192 FC? 03 

193 XEQ 08 

194 CLST 

195 RCL 01 

196 CHS 
197 RCL 00 

198 FC?C 01 

199 + 

200 FS? 03 

201 RTN 

202 GTO 08 
203▪LBL 03 

204 >"=?" 

205 PROMPT 

206▪LBL 10 

207 CF 00 

208 STO 00 

209 3 

210 / 

211 STO 01 

212 RCL 00 

213 * 

214 X<>Y 

215 ST- Y 

216 RCL 01 
217 * 

218 CHS 

219 R^ 

220 + 

221 2 

222 / 
223 RCL 01 

224 3 

225 Y^X 

226 + 

227 X<>Y 

228 CHS 
229 STO 02 

230 3 

231 / 

232 3 

233 Y^X 

234 RCL Y 

235 X^2 

236 + 

237 X<0? 

238 SF 00 

239 ABS 
240 SQRT 

241 X<>Y 

242 CHS 

243 FS?C 00 

244 GTO 00 

245 + 
246 . 

247 X<>Y 

248▪LBL 00 

249 3 

250 1/X 

251 STO 03 
252 RDN 

253 R-P 

254 RCL 03 

255 Y^X 

256 X<>Y 

257 RCL 03 

258 * 

259 X<>Y 

260 P-R 

261 STO 04 

262 X<>Y 
263 STO 05 

264 X<>Y 

265 RCL 02 

266 -3 

267 / 

268 . 
269 STO 10 

270 X<>Y 

271 R^ 

272 R^ 

273 R-P 

274 1/X 
275 X<>Y 

276 CHS 

277 X<>Y 

278 P-R 

279 XEQ 06 

280 STO 06 

281 X<>Y 

282 STO 07 

283 X<>Y 

284 RCL 05 

285 RCL 04 
286 XEQ 07 

287 STO 15 

288 FC? 04 

289 XEQ 05 

290 3 

291 SQRT 
292 .5 

293 * 

294 STO 09 

295 LASTX 

296 CHS 

297 STO 08 
298 RCL 05 

299 RCL 04 

300 XEQ 06 

301 STO 04 

302 X<>Y 

303 STO 05 

304 RCL 09 

305 RCL 08 

306 RCL 07 

307 RCL 06 

308 XEQ 06 
309 STO 06 

310 X<>Y 

311 STO 07 

312 X<>Y 

313 RCL 09 

314 RCL 08 
315 XEQ 06 

316 RCL 05 

317 RCL 04 

318 XEQ 07 

319 STO 16 

320 FC? 04 
321 XEQ 05 

322 RCL 09 

323 RCL 08 

324 RCL 05 

325 RCL 04 

326 XEQ 06 

327 RCL 07 

328 RCL 06 

329 XEQ 07 

330 STO 17 

331 FS? 04 
332 RTN 

333 GTO 05 

334▪LBL 06 

335 R-P 

336 R^ 

337 R^ 
338 R-P 

339 X<>Y 

340 RDN 

341 * 

342 RDN 

343 + 
344 R^ 

345 P-R 

346 RTN 

347▪LBL 07 

348 X<>Y 

349 RDN 

350 + 

351 RDN 

352 + 

353 R^ 

354 RTN 
355▪LBL 05 

356 RCL 01 

357 - 

358▪LBL 08 

359 ISG 10 

360 DEG 
361 CF 29 

362 FIX 07 

363 RND 

364 X<>Y 

365 RND 

366 FIX 00 
367 "X" 

368 ARCL 10 

369 >"=" 

370 FIX 02 

371 SF 29 

372 X<>Y 

373 X=0? 

374 SF 00 

375 X<>Y 

376 FS?C 00 

377 GTO 08 
378 ARCL Y 

379 X=0? 

380 GTO 09 

381 X>0? 

382 >"+" 

383▪LBL 08 
384 ARCL X 

385 X#0? 

386 >"I" 

387▪LBL 09 

388 FIX 05 

389 PROMPT 
390 END 

 

(542 bytes)

 

Registers, Labels and Flags 

REGISTERS COMMENTS  LABELS COMMENTS 

R00-R14 Work registers, general use  LBL00 Negative square root 

   LBL01 Not used 

   LBL02 Handle 2nd degrees (reused) 

   LBL03 Handle 3rd degrees (reused) 

   LBL04 Handle 4th degrees (reused) 

   LBL05 For 3rd degrees 

   LBL06 For 4th degrees 

   LBL07 Common general handlings 

   LBL08 Display of root values 

   LBL09-14 Process specific conditions 

  

FLAGS COMMENTS 

00 Indicate negative value before square root determination 

01 Indicate complex root 2nd degree 

02 Indicate complex root 4th degree 

03 Indicate complex root 3rd degree 

04 Indicate complex root 4th degree 

 

 

Downloads 

The RAW/TXT format of the program is available via the website: POLY234 (in zip file). 

 

https://hooxies.nl/?page_id=32202

